


Sekante durch Extremwerte

Dokumentnummer: DX1574
 Fachgebiet: Analysis, Polynome,
 Gerade, Gleichungssysteme
 Einsatz: 4HAK (drittes Lernjahr)



1 Aufgabe

Figure 1: Für eine Polynomfunktion dritten Grades ist die Gleichung ($y=kx+d$) der Sekante, die durch die beiden Extremwerte geht, zu bestimmen. SP ist der Schnittpunkt mit der y-Achse.



2 Lösung

2.1 Ablesbare Daten

```
(%i9) x1:-5$y1:0$
      x2:0$y2:2$
      x3:1$y3:0$
      x4:6$y4:0$
```

2.2 Verarbeitung

Ansatz

```
(%i17) g(x,y):=y=a*x**3+b*x**2+c*x+d;
(%o17) g(x,y):=y=a x^3 + b x^2 + c x + d
```

Gleichungssystem

```
(%i18) g1:g(x1,y1);
      g2:g(x2,y2);
      g3:g(x3,y3);
      g4:g(x4,y4);
(%o18) 0=d-5 c+25 b-125 a
(%o19) 2=d
(%o20) 0=d+c+b+a
(%o21) 0=d+6 c+36 b+216 a
```

Lösung

```
(%i22) l:solve([g1,g2,g3,g4],[a,b,c,d]);
(%o22) [[a=1/15,b=-2/15,c=-29/15,d=2]]
```

Funktion ermitteln

```
(%i23) Polynom:g(x,y),l;
(%o23)  $y = \frac{x^3}{15} - \frac{2x^2}{15} - \frac{29x}{15} + 2$ 
```

```
(%i24) f:rhs(Polynom);
(%o24)  $\frac{x^3}{15} - \frac{2x^2}{15} - \frac{29x}{15} + 2$ 
```

```
(%i25) f(x):='f;
(%o25)  $f(x) := \frac{x^3}{15} - \frac{2x^2}{15} - \frac{29x}{15} + 2$ 
```

Extremwerte ermitteln

```
(%i26) ab:diff(f(x),x);
(%o26)  $\frac{x^2}{5} - \frac{4x}{15} - \frac{29}{15}$ 
```

```
(%i28) l:realroots(ab),numer;
(%o28) [x=-2.513130694627762,x=3.84646400809288]
```

```
(%i33) x1:ev(x,l[1]);y1:f(x1);
      x2:ev(x,l[2]);y2:f(x2);
(%o33) -2.513130694627762
(%o34) 4.958442831058867
(%o35) 3.84646400809288
(%o36) -3.615232954515657
```

Sekante bestimmen

```
(%i37) g(x,y):=y=k*x+d;
(%o37)  $g(x,y) := y = kx + d$ 
```

```
(%i38) g1:g(x1,y1);  
        g2:g(x2,y2);  
(%o38) 4.958442831058867=d-2.513130694627762 k  
(%o39) -3.615232954515657=3.84646400809288 k+d  
  
(%i42) l:algsys([g1,g2],[k,d]),numer;  
(%o42) [[k=-1.348148217421314,d=1.570370365928742]]
```

Steigung der Sekante

```
(%i47) k:ev(k,l);  
        k:floor(k*100+0.5)/100.0;  
(%o47) -1.348148217421314  
(%o48) -1.35
```

Sekante: Abschnitt auf der y-Achse

```
(%i50) d:ev(d,l);  
        d:floor(d*100+0.5)/100.0;  
(%o50) 1.570370365928742  
(%o51) 1.57
```

2.3 Ausgabe

```
(%i54) Sekante:y=k*x+d;  
(%o54) y=1.57-1.35 x
```

Das ist die gesuchte Sekantengleichung!